
• Hardware description language (HDL): 
– specifies logic function only
– Computer-aided design (CAD) tool produces or 

synthesizes the optimized gates

• Most commercial designs built using HDLs
• Two leading HDLs:

– SystemVerilog
• developed in 1984 by Gateway Design Automation
• IEEE standard (1364) in 1995
• Extended in 2005 (IEEE STD 1800-2009)

– VHDL 2008
• Developed in 1981 by the Department of Defense
• IEEE standard (1076) in 1987
• Updated in 2008 (IEEE STD 1076-2008)

Introduction

• Simulation
– Inputs applied to circuit

– Outputs checked for correctness

– Millions of dollars saved by debugging in simulation instead of 
hardware

– Example: correcting a mistake in a cutting-edge integrated circuit costs more than a million dollars and 
takes several months. Intel’s infamous FDIV (floating point division) bug in the Pentium processor 
forced the company to recall chips after they had shipped, at a total cost of $475 million. 

– Logic simulation is essential to test a system before it is built.

• Synthesis
– Transforms HDL code into a netlist describing the hardware (i.e., a 

list of gates and the wires connecting them)

HDL to Gates

• Simulation
– Inputs applied to circuit

– Outputs checked for correctness

– Millions of dollars saved by debugging in simulation 
instead of hardware

• Synthesis
– Transforms HDL code into a netlist describing the hardware 

(i.e., a list of gates and the wires connecting them)

IMPORTANT: When using an HDL, think of the hardware the HDL 
should produce

HDL to Gates

• In our experience, the best way to learn an HDL is by example. 
HDLs have specific ways of describing various classes of logic; 
these ways are called idioms.

• This chapter will teach you how to write the proper HDL idioms for 
each type of block and then how to put the blocks together to 
produce a working system. 

• When you need to describe a particular kind of hardware, look for 
a similar example and adapt it to your purpose.

• We do not attempt to rigorously define all the syntax of the HDLs, 
because that is deathly boring and because it tends to encourage 
thinking of HDLs as programming languages, not shorthand for 
hardware.

HDL

a
b y
c

Verilog

Module

Two types of Modules:
– Behavioral: describe what a module does

– Structural: describe how it is built from simpler 
modules

SystemVerilog Modules

module example(input  logic a, b, c,

output logic y);

assign y = ~a & ~b & ~c | a & ~b & ~c | a & ~b &  c;

endmodule

SystemVerilog:

Behavioral SystemVerilog

module example(input  logic a, b, c,

output logic y);

assign y = ~a & ~b & ~c | a & ~b & ~c | a & ~b &  c;

endmodule

SystemVerilog:

Behavioral SystemVerilog

• module/endmodule:  required to begin/end module
• example:  name of the module

• Operators:
~:  NOT
&:  AND
|:  OR

HDL Simulation

module example(input  logic a, b, c,

output logic y);

assign y = ~a & ~b & ~c | a & ~b & ~c | a & ~b &  c;

endmodule

SystemVerilog:

un5_y

un8_y

y

yc

b

a

HDL Synthesis

module example(input  logic a, b, c,

output logic y);

assign y = ~a & ~b & ~c | a & ~b & ~c | a & ~b &  c;

endmodule

SystemVerilog:

Synthesis:

• Case sensitive
– Example: reset and Reset are not the same signal.

• No names that start with numbers 
– Example: 2mux is an invalid name

• Whitespace ignored

• Comments:
– // single line comment

– /* multiline

comment */

SystemVerilog Syntax

module and3(input  logic a, b, c,

output logic y);

assign y = a & b & c;

endmodule

module inv(input  logic a,

output logic y);

assign y = ~a;

endmodule

module nand3(input  logic a, b, c

output logic y);

logic n1;                   // internal signal

and3 andgate(a, b, c, n1);  // instance of and3

inv inverter(n1, y);       // instance of inv

endmodule

Structural Modeling - Hierarchy

module gates(input  logic [3:0]  a, b,

output logic [3:0] y1, y2, y3, y4, y5);

/* Five different two-input logic 

gates acting on 4 bit busses */

assign y1 = a & b;    // AND

assign y2 = a | b;    // OR

assign y3 = a ^ b;    // XOR

assign y4 = ~(a & b); // NAND

assign y5 = ~(a | b); // NOR

endmodule

// single line comment

/*…*/ multiline comment 

Bitwise Operators

module and8(input  logic [7:0] a, 

output logic       y);

assign y = &a;

// &a is much easier to write than

// assign y = a[7] & a[6] & a[5] & a[4] &

//            a[3] & a[2] & a[1] & a[0];

endmodule

Reduction Operators
Reduction operators imply a multiple-input gate acting on a single bus.

module mux2(input  logic [3:0] d0, d1, 

input  logic s,

output logic [3:0] y);

assign y = s ? d1 : d0; 

endmodule

? : is also called a ternary operator because it   

operates on 3 inputs: s, d1, and d0.

Conditional Assignment
Conditional assignments select the output from among alternatives based on an input called 
the condition.

module fulladder(input  logic a, b, cin, 

output logic s, cout);

logic p, g;   // internal nodes

assign p = a ^ b;

assign g = a & b;

assign s = p ^ cin;

assign cout = g | (p & cin);

endmodule

p

g s

un1_cout cout

cout

s

cin

b

a

Internal Variables

~ NOT

*, /, % mult, div, mod

+, - add,sub

<<, >> shift

<<<, >>> arithmetic shift

<, <=, >, >= comparison

==, != equal, not equal

&, ~& AND, NAND

^, ~^ XOR, XNOR

|, ~| OR, NOR

?: ternary operator

Highest

Lowest

Precedence



Chapter 4 <29> Digital Design and Computer Architecture: ARM® Edition © 2015



Chapter 4 <30> Digital Design and Computer Architecture: ARM® Edition © 2015

Number # Bits Base Decimal 

Equivalent

Stored

3'b101 3 binary 5 101

'b11 unsized binary 3 00…0011

8'b11 8 binary 3 00000011

8'b1010_1011 8 binary 171 10101011

3'd6 3 decimal 6 110

6'o42 6 octal 34 100010

8'hAB 8 hexadecimal 171 10101011

42 Unsized decimal 42 00…0101010

Format: N'Bvalue
N = number of bits, B = base

N'B is optional but recommended (default is decimal)

Numbers

y_1[3:0]

y[3:0]
[3:0]

en

a[3:0]
[3:0] [3:0][3:0]

module tristate(input  logic [3:0] a, 

input  logic en, 

output tri   [3:0] y);

assign y = en ? a : 4'bz;

endmodule

Z: Floating Output

SystemVerilog:

assign y = {a[2:1], {3{b[0]}}, a[0], 6'b100_010};

// if y is a 12-bit signal, the above statement produces:

y = a[2] a[1] b[0] b[0] b[0] a[0] 1 0 0 0 1 0

// underscores (_) are used for formatting only to make

// it easier to read. SystemVerilog ignores them. 

Bit Manipulations: Example 1

Often it is necessary to operate on a subset of a bus or to concatenate (join 
together) signals to form busses. These operations are collectively
known as bit swizzling. In HDL Example 4.12, y is given the 9-bit value 
c2c1d0d0d0c0101 using bit swizzling operations

module mux2_8(input  logic [7:0] d0, d1,

input  logic       s,

output logic [7:0] y);

mux2 lsbmux(d0[3:0], d1[3:0], s, y[3:0]);

mux2 msbmux(d0[7:4], d1[7:4], s, y[7:4]);

endmodule
mux2

lsbmux

mux2

msbmux

y[7:0]
[7:0]

s

d1[7:0]
[7:0]

d0[7:0]
[7:0]

s

[3:0]
d0[3:0]

[3:0]
d1[3:0]

[3:0]
y[3:0]

s

[7:4]
d0[3:0]

[7:4]
d1[3:0]

[7:4]
y[3:0]

Bit Manipulations: Example 2
SystemVerilog:

module example(input  logic a, b, c,

output logic y);

logic ab, bb, cb, n1, n2, n3;

assign #1 {ab, bb, cb} = ~{a, b, c};

assign #2 n1 = ab & bb & cb;

assign #2 n2 = a & bb & cb;

assign #2 n3 = a & bb & c;

assign #4 y = n1 | n2 | n3;

endmodule

Delays

module example(input  logic a, b, c,

output logic y);

logic ab, bb, cb, n1, n2, n3;

assign #1 {ab, bb, cb} = 

~{a, b, c};

assign #2 n1 = ab & bb & cb;

assign #2 n2 = a & bb & cb;

assign #2 n3 = a & bb & c;

assign #4 y = n1 | n2 | n3;

endmodule

Delays

module example(input  logic a, b, c,

output logic y);

logic ab, bb, cb, n1, n2, n3;

assign #1 {ab, bb, cb} = 

~{a, b, c};

assign #2 n1 = ab & bb & cb;

assign #2 n2 = a & bb & cb;

assign #2 n3 = a & bb & c;

assign #4 y = n1 | n2 | n3;

endmodule

Delays
1

module example(input  logic a, b, c,

output logic y);

logic ab, bb, cb, n1, n2, n3;

assign #1 {ab, bb, cb} = 

~{a, b, c};

assign #2 n1 = ab & bb & cb;

assign #2 n2 = a & bb & cb;

assign #2 n3 = a & bb & c;

assign #4 y = n1 | n2 | n3;

endmodule

Delays
2

module example(input  logic a, b, c,

output logic y);

logic ab, bb, cb, n1, n2, n3;

assign #1 {ab, bb, cb} = 

~{a, b, c};

assign #2 n1 = ab & bb & cb;

assign #2 n2 = a & bb & cb;

assign #2 n3 = a & bb & c;

assign #4 y = n1 | n2 | n3;

endmodule

Delays

2

module example(input  logic a, b, c,

output logic y);

logic ab, bb, cb, n1, n2, n3;

assign #1 {ab, bb, cb} = 

~{a, b, c};

assign #2 n1 = ab & bb & cb;

assign #2 n2 = a & bb & cb;

assign #2 n3 = a & bb & c;

assign #4 y = n1 | n2 | n3;

endmodule

Delays

4

STRUCTURAL MODELING

• The previous section discussed behavioral modeling, describing a module in terms 
of the relationships between inputs and outputs. 

• This section examines structural modeling, describing a module in terms of how it 
is composed of simpler modules.

• For example, HDL Example 4.14 shows how to assemble a 4:1 multiplexer from 
three 2:1 multiplexers

HDL Example 4.15 uses structural modeling to construct a 2:1 multiplexer from a pair of 
tristate buffers. Building logic out of tristates is not recommended, however.

• HDL Example 4.16 shows how modules can access part of a bus. An 8-bit wide 2:1 
multiplexer is built using two of the 4-bit 2:1 multiplexers already defined, operating on the 
low and high nibbles of the byte.

• In general, complex systems are designed hierarchically. The overall system is described 
structurally by instantiating its major components. Each of these components is described 
structurally from its building blocks, and so forth recursively until the pieces are simple 
enough to describe behaviorally. It is good style to avoid (or at least to minimize) mixing 
structural and behavioral descriptions within a single module.

• SystemVerilog uses idioms to describe 
latches, flip-flops and FSMs

• HDL synthesizers recognize certain idioms 
and turn them into specific sequential 
circuits.

• Other coding styles may simulate correctly 
but produce incorrect hardware

Sequential Logic

module flop(input  logic       clk, 

input  logic [3:0] d, 

output logic [3:0] q);

always_ff @(posedge clk)

q <= d;                // pronounced “q gets d”

endmodule

D Flip-Flop
The vast majority of modern commercial systems are built with registers
using positive edge-triggered D flip-flops. HDL Example 4.17 shows the
idiom for such flip-flops.

General Structure:

always @(sensitivity list)

statement;

Whenever the event in sensitivity list occurs, statement is 
executed

Always Statement

In SystemVerilog always statements and VHDL process statements, signals keep their old value until an event in 
the sensitivity list takes place that explicitly causes them to change. Hence, such code, with appropriate 
sensitivity lists, can be used to describe sequential circuits with memory. For example, the flip-flop includes 
only clk in the sensitive list. It remembers its old value of q until the next rising edge of the clk, even if d 
changes in the interim. 

In contrast, SystemVerilog continuous assignment statements (assign) and VHDL concurrent assignment 
statements (<=) are reevaluated anytime any of the inputs on the right hand side changes. Therefore, such 
code necessarily describes combinational logic.

module flopr(input  logic       clk,

input  logic       reset, 

input  logic [3:0] d, 

output logic [3:0] q);

// synchronous reset

always_ff @(posedge clk)

if (reset) q <= 4'b0;

else       q <= d;

endmodule

q[3:0]

q[3:0]
[3:0]

d[3:0]
[3:0]

reset

clk
[3:0]

Q[3:0]
[3:0]

D[3:0]
R

Resettable D Flip-Flop
When simulation begins or 
power is first applied to a 
circuit, the output of
a flop or register is unknown. 
This is indicated with x in 
SystemVerilog and
u in VHDL. Generally, it is 
good practice to use 
resettable registers so that
on powerup you can put your 
system in a known state. The 
reset may be
either asynchronous or 
synchronous. Recall that 
asynchronous reset
occurs immediately, whereas 
synchronous reset clears the 
output only on the next rising 
edge of the clock.

module flopr(input  logic       clk,

input  logic       reset, 

input  logic [3:0] d, 

output logic [3:0] q);

// asynchronous reset

always_ff @(posedge clk, posedge reset)

if (reset) q <= 4'b0;

else       q <= d;

endmodule

q[3:0]

R

q[3:0]
[3:0]

d[3:0]
[3:0]

reset

clk
[3:0]

Q[3:0]
[3:0]

D[3:0]

Resettable D Flip-Flop

module flopren(input  logic       clk,

input  logic       reset, 

input  logic       en, 

input  logic [3:0] d, 

output logic [3:0] q);

// enable and asynchronous reset

always_ff @(posedge clk, posedge reset)

if      (reset) q <= 4'b0;

else if (en)    q <= d;

endmodule

D Flip-Flop with Enable
Enabled registers respond to 
the clock only when the enable 
is asserted.
HDL Example 4.19 shows an 
asynchronously resettable 
enabled register
that retains its old value if 
both reset and en are FALSE.

module latch(input  logic       clk, 

input  logic [3:0] d, 

output logic [3:0] q);

always_latch

if (clk) q <= d;

endmodule

Warning: We don’t use latches in this text. But you might write code that 
inadvertently implies a latch. Check synthesized hardware – if it has latches
in it, there’s an error.

lat

q[3:0]

q[3:0]
[3:0]d[3:0]

[3:0]

clk

[3:0]
D[3:0] [3:0]

Q[3:0]
C

Latch
Recall from Section 3.2.2 that a D 
latch is transparent when the clock is 
HIGH, allowing data to flow from 
input to output. The latch becomes
opaque when the clock is LOW, 
retaining its old state. HDL Example 
4.21 shows the idiom for a D latch.

In Section 4.2, we used assignment statements to describe combinational  logic behaviorally. SystemVerilog
always statements and VHDL process statements are used to describe sequential circuits, because they 
remember the old state when no new state is prescribed. However, always/process statements can also be used 
to describe combinational logic behaviorally if the sensitivity list is written to respond to changes in all of the 
inputs and the body prescribes the output value for every possible input combination. HDL Example 4.22 uses 
always/process statements to describe a bank of four inverters (see Figure 4.3 for the synthesized circuit).

HDLs support blocking and nonblocking assignments in an always/process statement. A group of blocking 
assignments are evaluated in the order in which they appear in the code, just as one would expect in a standard
programming language. A group of nonblocking assignments are  evaluated concurrently; all of the statements 
are evaluated before any of the signals on the left hand sides are updated.

• Statements that must be inside always
statements:
– if / else

– case, casez

Other Behavioral Statements

// combinational logic using an always statement

module gates(input  logic [3:0] a, b, 

output logic [3:0] y1, y2, y3, y4, y5);

always_comb // need begin/end because there is

begin            // more than one statement in always

y1 = a & b;    // AND

y2 = a | b;    // OR

y3 = a ^ b;    // XOR

y4 = ~(a & b); // NAND

y5 = ~(a | b); // NOR

end

endmodule

This hardware could be described with assign statements using fewer lines 
of code, so it’s better to use assign statements in this case.

Combinational Logic using always

module sevenseg(input  logic [3:0] data, 

output logic [6:0] segments);

always_comb

case (data)

//                     abc_defg

0: segments =       7'b111_1110;

1: segments =       7'b011_0000;

2: segments =       7'b110_1101;

3: segments =       7'b111_1001;

4: segments =       7'b011_0011;

5: segments =       7'b101_1011;

6: segments =       7'b101_1111;

7: segments =       7'b111_0000;

8: segments =       7'b111_1111;

9: segments =       7'b111_0011;

default: segments = 7'b000_0000; // required

endcase

endmodule

Combinational Logic using case



Chapter 4 <65> Digital Design and Computer Architecture: ARM® Edition © 2015

• case statement  implies combinational logic 
only if all possible input combinations described

• Remember to use default statement

Combinational Logic using case

module priority_casez(input  logic [3:0] a, 

output logic [3:0] y);

always_comb

casez(a)

4'b1???: y = 4'b1000; // ?=don’t care

4'b01??: y = 4'b0100;

4'b001?: y = 4'b0010;

4'b0001: y = 4'b0001;

default: y = 4'b0000;

endcase

endmodule

Combinational Logic using casez

The casez statement acts like a case statement 
except that it also recognizes ? as don’t care.

Blocking and 
Nonblocking 
Assignments

• The guidelines on the next page  
explain when and how to use each 
type of assignment. 

• If these guidelines are not followed, it 
is possible to write code that appears 

– to work in simulation 

– but synthesizes to incorrect 
hardware.

• <= is nonblocking assignment
– Occurs simultaneously with others

• = is blocking assignment
– Occurs in order it appears in file
// Good synchronizer using 

// nonblocking assignments

module syncgood(input  logic clk,

input  logic d,

output logic q);

logic n1;

always_ff @(posedge clk)

begin

n1 <= d;  // nonblocking

q  <= n1; // nonblocking

end

endmodule

// Bad synchronizer using 

// blocking assignments

module syncbad(input logic  clk,

input  logic d,

output logic q);

logic n1;

always_ff @(posedge clk)

begin

n1 = d;  // blocking

q  = n1; // blocking

end

endmodule

Blocking vs. Nonblocking Assignment

• Synchronous sequential logic: use always_ff @(posedge

clk)and nonblocking assignments (<=)
always_ff @ (posedge clk)

q <= d; // nonblocking

• Simple combinational logic: use continuous assignments 
(assign…)

assign y = a & b; 

• More complicated combinational logic: use always_comb
and blocking assignments (=)

• Assign a signal in only one always statement or continuous 
assignment statement.

Rules for Signal Assignment

CLK
M Nk knext

state

logic

output

logic
inputs outputs

state
next

state

• Recall that a finite state machine (FSM) consists of a state register and two blocks of 
combinational logic to compute the next state and the output given the current state 
and the input. HDL descriptions of state machines are correspondingly divided into 
three parts to model the state register, the next state logic, and the output logic.

• Three blocks:

– next state logic

– state register

– output logic

Finite State Machines (FSMs)



Divide-by-3 Counter

A divide-by-N counter has one output and no inputs. The output Y is HIGH for
one clock cycle out of every N. In other words, the output divides the frequency
of the clock by N. The waveform and state transition diagram for a divide-by-3
counter is shown in Figure 3.28. Sketch circuit designs for such a counter using
binary and one-hot state encodings.

The double circle indicates the reset state

S 0

S 1

S 2

FSM Example: Divide by 3

module divideby3FSM (input  logic clk, 

input  logic reset, 

output logic q);

typedef enum logic [1:0] {S0, S1, S2} statetype;

statetype [1:0] state, nextstate;  

// state register

always_ff @ (posedge clk, posedge reset)

if (reset) state <= S0;

else       state <= nextstate;

// next state logic

always_comb

case (state)

S0:      nextstate = S1;

S1:      nextstate = S2;

S2:      nextstate = S0;

default: nextstate = S0;

endcase

// output logic

assign q = (state == S0);

endmodule

FSM in SystemVerilog

Verilog

• So far all of our modules have had fixed-width 
inputs and outputs. 

• For example, we had to define separate modules for 
4- and 8-bit wide 2:1 multiplexers.

• HDLs permit variable bit widths using parameterized 
modules.

• HDL Example 4.34 declares a parameterized 2:1 
multiplexer with a default width of 8, then uses it to 
create 8- and 12-bit 4:1 multiplexers.

Parameterized Modules

2:1 mux:
module mux2

#(parameter width = 8)  // name and default value

(input  logic [width-1:0] d0, d1, 

input  logic             s,

output logic [width-1:0] y);

assign y = s ? d1 : d0; 

endmodule

Instance with 8-bit bus width (uses default):
mux2 myMux(d0, d1, s, out);

Instance with 12-bit bus width:
mux2 #(12) lowmux(d0, d1, s, out); 

Parameterized Modules

• A testbench is an HDL module that is used to test another 
module, called the device under test (DUT). 

• The testbench contains statements to apply inputs to the DUT 
and, ideally, to check that the correct outputs are produced. 

• The input and desired output patterns are called test vectors.
• It instantiates the DUT, then applies the inputs. Blocking 

assignments and delays are used to apply the inputs in the 
appropriate order. 

• The user must view the results of the simulation and verify by 
inspection that the correct outputs are produced. 

• Testbenches are simulated the same as other HDL modules. 
However, they are not synthesizeable.

Testbenches

• Types:

– Simple

– Self-checking

– Self-checking with testvectors

Testbenches Types

• Write SystemVerilog code to implement the 
following function in hardware: 

y = bc + ab

• Name the module sillyfunction

Testbench Example

• Write SystemVerilog code to implement the 
following function in hardware: 

y = bc + ab

module sillyfunction(input  logic a, b, c, 

output logic y);

assign y = ~b & ~c | a & ~b;

endmodule

Testbench Example

module testbench1();

logic a, b, c;

logic y;

// instantiate device under test

sillyfunction dut(a, b, c, y);

// apply inputs one at a time and wait 10 time units

initial begin

a = 0; b = 0; c = 0; #10;

c = 1; #10;

b = 1; c = 0; #10;

c = 1; #10;

a = 1; b = 0; c = 0; #10;

c = 1; #10;

b = 1; c = 0; #10;

c = 1; #10;

end

endmodule

Simple Testbench

Simple Testbenches

• A simple testbench instantiates the DUT, 
then applies the inputs.

• Blocking assignments and delays are used 
to apply the inputs in the appropriate 
order. 

• The user must view the results of the 
simulation and verify by inspection that 
the correct outputs are produced.

• Checking for correct outputs is tedious and error-prone.
Moreover, determining the correct outputs is much easier 
when the design is fresh in your mind; if you make minor 
changes and need to retest weeks later, determining the 
correct outputs becomes a hassle. 

• A much better approach is to write a self-checking testbench.

Self-checking Testbenches

module testbench2();

logic  a, b, c;

logic y;

sillyfunction dut(a, b, c, y);  // instantiate dut

initial begin // apply inputs, check results one at a 

time

a = 0; b = 0; c = 0; #10;

if (y !== 1) $display("000 failed.");

c = 1; #10;

if (y !== 0) $display("001 failed.");

b = 1; c = 0; #10;

if (y !== 0) $display("010 failed.");

c = 1; #10;

if (y !== 0) $display("011 failed.");

a = 1; b = 0; c = 0; #10;

if (y !== 1) $display("100 failed.");

c = 1; #10;

if (y !== 1) $display("101 failed.");

b = 1; c = 0; #10;

if (y !== 0) $display("110 failed.");

c = 1; #10;

if (y !== 0) $display("111 failed.");

end

endmodule

Self-checking Testbench

module sillyfunction(input  logic a, b, c, 

output logic y);

assign y = ~b & ~c | a & ~b;

endmodule

• Writing code for each test vector also becomes tedious, 
especially for modules that require a large number of vectors. 
An even better approach  is to place the test vectors in a 
separate file. 

• The testbench simply reads the test vectors from the file, 
applies the input test vector to the DUT, waits, checks that the 
output values from the DUT match the output vector, and 
repeats until reaching the end of the test vectors file.

• Testvector file: inputs and expected outputs
• Testbench:

1. Generate clock for assigning inputs, reading outputs
2. Read testvectors file into array
3. Assign inputs, expected outputs
4. Compare outputs with expected outputs and report errors

Testbench with Testvectors

• Testbench clock: 
– assign inputs (on rising edge)

– compare outputs with expected outputs (on falling 
edge).

• Testbench clock also used as clock for synchronous 
sequential circuits

Assign

Inputs

Compare

Outputs to

Expected

CLK

Testbench with Testvectors

• File: example.tv (a text file 
containing the inputs and expected 

output written in binary)

• contains vectors of abc_y expected

000_1

001_0

010_0

011_0

100_1

101_1

110_0

111_0

Testvectors File

module testbench3();

logic        clk, reset;

logic        a, b, c, yexpected;

logic        y;

logic [31:0] vectornum, errors;    // bookkeeping variables

logic [3:0]  testvectors[10000:0]; // array of testvectors

// instantiate device under test

sillyfunction dut(a, b, c, y);

// generate clock

always     // no sensitivity list, so it always executes

begin

clk = 1; #5; clk = 0; #5;

end

1. Generate Clock

// at start of test, load vectors and pulse reset

initial

begin

$readmemb("example.tv", testvectors);

vectornum = 0; errors = 0;

reset = 1; #27; reset = 0;

end

// Note: $readmemh reads testvector files written in

// hexadecimal

2. Read Testvectors into Array

// apply test vectors on rising edge of clk

always @(posedge clk)

begin

#1; {a, b, c, yexpected} = testvectors[vectornum];

end

3. Assign Inputs & Expected Outputs

// check results on falling edge of clk

always @(negedge clk)

if (~reset) begin // skip during reset

if (y !== yexpected) begin  

$display("Error: inputs = %b", {a, b, c});

$display("  outputs = %b (%b expected)",y,yexpected);

errors = errors + 1;

end

// Note: to print in hexadecimal, use %h. For example,

//       $display(“Error: inputs = %h”, {a, b, c});

4. Compare with Expected Outputs

// increment array index and read next testvector

vectornum = vectornum + 1;

if (testvectors[vectornum] === 4'bx) begin 

$display("%d tests completed with %d errors", 

vectornum, errors);

$finish;

end

end

endmodule

// === and !== can compare values that are 1, 0, x, or z.

4. Compare with Expected Outputs

• Hardware description languages (HDLs) are extremely important tools for modern 
digital designers. Once you have learned SystemVerilog or VHDL, you will be able to 
specify digital systems much faster than if you had to draw the complete 
schematics. 

• The debug cycle is also often much faster, because modifications require code 
changes instead of tedious schematic rewiring. However, the debug cycle can be 
much longer using HDLs if you don’t have a good idea of the hardware your code 
implies.

• HDLs are used for both simulation and synthesis. Logic simulation is a powerful way 
to test a system on a computer before it is turned into hardware. Simulators let you 
check the values of signals inside your system that might be impossible to measure 
on a physical piece of hardware.

• Logic synthesis converts the HDL code into digital logic circuits.
• The most important thing to remember when you are writing HDL code is that you 

are describing real hardware, not writing a computer program.
• The most common beginner’s mistake is to write HDL code without
thinking about the hardware you intend to produce.

Summary


